# Maintainer: Jaroslav Lichtblau <svetlemodry@archlinux.org>
# Contributor: Stéphane Gaudreault <stephane@archlinux.org>
# Contributor: Thomas Dziedzic < gostrc at gmail >
# Contributor: Thomas Baechler <thomas@archlinux.org>

pkgname=ginac
pkgver=1.8.6
pkgrel=1
pkgdesc="C++ library for symbolic calculations"
arch=('loong64' 'x86_64')
url='https://www.ginac.de/'
license=('GPL')
depends=('cln')
makedepends=('fig2dev' 'doxygen' 'texlive-bin' 'python')
source=(https://www.ginac.de/$pkgname-$pkgver.tar.bz2)
sha256sums=('00b320b1116cae5b7b43364dbffb7912471d171f484d82764605d715858d975b')

build() {
  cd "${srcdir}"/$pkgname-$pkgver

  ./configure --prefix=/usr --libexecdir=/usr/lib/ginac --disable-static
  make
}

check() {
  cd "${srcdir}"/$pkgname-$pkgver

  make check
}

package() {
  cd "${srcdir}"/$pkgname-$pkgver

  make DESTDIR="${pkgdir}" install
}
